SchriftartenÜbersicht per Bericht ausgeben Teil 1
Versionen: Access 2007, 2003, 2002/XP, 2000 und 97

Bei der Entwicklung Ihrer Datenbanken ist auch immer die Frage nach der für Formulare und Berichte zu verwendenden Schriftart zu klären. Die Wahl hängt letztendlich davon ab, welche Schriftarten auf dem jeweiligen System verfügbar sind. Um die Auswahl zu erleichtern, wäre ein kleiner Katalog mit Mustern der verfügbaren Schriftarten recht praktisch. Dazu würde sich ein Bericht für die Ausgabe der Schriftartenmuster anbieten. Allerdings kann dieser Bericht normalerweise nur statisch aufgebaut sein: Für jede Schriftart müsste ein Bezeichnungsfeld mit Mustertext angelegt und dessen Eigenschaft "Schriftart" entsprechend eingestellt sein. Bei Systemen, auf denen mitunter einige Hundert Schriftarten installiert sind, ist das jedoch nicht praktikabel. Nutzen Sie stattdessen eine Lösung, die den Bericht und seine Datenbasis beim Öffnen so modifiziert, dass automatisch die auf dem System verfügbaren Schriftarten ausgegeben werden.

Basis einer solchen Lösung ist eine Tabelle "tblFontList" mit den Feldern "Mustertext" und "Schriftart". Diese Tabelle liegt einem Bericht "Schriftarten-Übersicht" zugrunde, in dessen Detailbereich die Felder "Schriftart" und "Mustertext" enthalten sind. Beim Öffnen des Berichtes wird zunächst eine Funktion "FillFontTable()" aufgerufen, die über API-Funktionen die Namen der aktuell auf dem System verfügbaren Schriftarten ermittelt und pro verfügbarer Schriftart den Namen nebst einem Mustertext in die Tabelle "tblFontList" schreibt. Die Datenbasis für den Bericht bilden dann anschließend genau so viele Datensätze, wie Schriftarten auf dem System vorhanden sind. Bei der Ausgabe des Berichtes muss jetzt nur noch dafür gesorgt werden, dass das Feld "Mustertext" in der dazugehörigen Schriftart formatiert wird. Dies lässt sich über eine entsprechende Ereignisprozedur des Berichtes erledigen. Mehr dazu später, treffen wir nun zunächst die Vorbereitungen für den Einsatz der Lösung in Ihren Datenbanken:

· Legen Sie eine Tabelle "tblFontList" mit den folgenden Feldern an:

Schriftart, Datentyp "Text", Feldlänge 100
Mustertext, Datentyp "Text", Feldlänge 250

Ein Primärschlüsselfeld muss nicht gesetzt werden.

· Legen Sie ein neues Modul an und benennen Sie es "modFontList" oder ähnlich.

· Kopieren Sie die folgenden Deklarationen und Funktionen in das Modul:

Option Explicit

Private Const LF_FACESIZE = 32

Private Type LOGFONT
  lfHeight As Long
  lfWidth As Long
  lfEscapement As Long
  lfOrientation As Long
  lfWeight As Long
  lfItalic As Byte
  lfUnderline As Byte
  lfStrikeOut As Byte
  lfCharSet As Byte
  lfOutPrecision As Byte
  lfClipPrecision As Byte
  lfQuality As Byte
  lfPitchAndFamily As Byte
  lfFaceName(LF_FACESIZE) As Byte
End Type

Private Type NEWTEXTMETRIC
  tmHeight As Long
  tmAscent As Long
  tmDescent As Long
  tmInternalLeading As Long
  tmExternalLeading As Long
  tmAveCharWidth As Long
  tmMaxCharWidth As Long
  tmWeight As Long
  tmOverhang As Long
  tmDigitizedAspectX As Long
  tmDigitizedAspectY As Long
  tmFirstChar As Byte
  tmLastChar As Byte
  tmDefaultChar As Byte
  tmBreakChar As Byte
  tmItalic As Byte
  tmUnderlined As Byte
  tmStruckOut As Byte
  tmPitchAndFamily As Byte
  tmCharSet As Byte
  ntmFlags As Long
  ntmSizeEM As Long
  ntmCellHeight As Long
  ntmAveWidth As Long
End Type

Declare Function EnumFontFamilies Lib "gdi32" _
        Alias "EnumFontFamiliesA" _
       (ByVal hDC As Long, _
        ByVal lpszFamily As String, _
        ByVal lpEnumFontFamProc As Long, _
        LParam As Any) As Long

Private Declare Function GetDC Lib "user32" _
        (ByVal hWnd As Long) As Long

Private Declare Function ReleaseDC Lib "user32" _
        (ByVal hWnd As Long, _
         ByVal hDC As Long) As Long

Public arrFonts() As String
Public intNumFonts As Integer

'----------------------------------------------

Function FillFontTable() As Integer
  Dim db As DAO.Database
  Dim rs As DAO.Recordset
  Dim R As Variant, I As Integer

  FillFontTable = 0
  DoCmd.SetWarnings False
  DoCmd.RunSQL "delete * from tblFontList", False
  DoCmd.SetWarnings True

  R = APIFontList()
  If R = 0 Then
    Beep
    MsgBox "Schriftarten konnten nicht " & _
           "ausgelesen werden..."
    Exit Function
  Else
    FillFontTable = R
  End If

  Set db = CurrentDb()
  Set rs = db.OpenRecordset("tblFontList", _
                            dbOpenDynaset)
  For I = 1 To R
    rs.AddNew
    rs("Mustertext") = "Franz jagt im komplett" & _
       " verwahrlosten Taxi quer durch Bayern " & _
       "- äöü ÄÖÜ ß € - 1234567890"
    rs("Schriftart") = Trim$(arrFonts(I))
    rs.Update
  Next I
  rs.Close
  DoEvents

End Function

Public Function APIFontList() As Integer
  Dim R As Variant, hDC As Long, I As Integer

  intNumFonts = 0
  Erase arrFonts()
  hDC = GetDC(Application.hWndAccessApp)
  EnumFontFamilies hDC, _
                   vbNullString, _
                   AddressOf EnumFontFamProc, 0&
  ReleaseDC Application.hWndAccessApp, hDC
  APIFontList = intNumFonts

End Function

Function EnumFontFamProc(lpNLF As LOGFONT, _
                         lpNTM As NEWTEXTMETRIC, _
                         ByVal FontType As Long, _
                         LParam As Long) As Long
  Dim strAPIName As String
  Dim strFontname As String

  On Error Resume Next
  strAPIName = StrConv(lpNLF.lfFaceName, vbUnicode)
  strFontname = CVBStr(strAPIName)
  intNumFonts = intNumFonts + 1
  ReDim Preserve arrFonts(intNumFonts)
  arrFonts(intNumFonts) = strFontname
  EnumFontFamProc = 1

End Function

Function CVBStr(varString As Variant) As String

  If InStr(varString, vbNullChar) <> 0 Then
    CVBStr = Left$(varString, _
             InStr(varString, vbNullChar) - 1)
  Else
    CVBStr = varString
  End If

End Function
· Speichern Sie das Modul.

Damit sind zunächst die Grundlagen für die Schriftarten-Übersicht bereitgestellt. Im zweiten Teil dieses Tipps der Woche in der nächsten Ausgabe von Access Weekly erläutern wir dann den Aufbau des dazugehörenden Berichtes, erklären das Zusammenspiel von Bericht und Modul und wie die Lösung VBA-technisch funktioniert.

